
Abstract

Terabytes of data files are created as a result of the D0 experiment at the Fermi National Accelerator Laboratory. Physicists run analysis programs on collections of data files, each of these activities is referred to as an analysis project. Only a percentage of the analysis projects run by the D0 collaborators contain shared data files. The more frequently used data files are stored in caches for fast access by the SAM (Sequential Access via Meta-data) system, the automatic data file distribution system developed for the D0 experiment. In order to better understand the amount of cache space needed by SAM, it is necessary to understand the access patterns of the files (e.g. the amount of files shared by the analysis projects).  This paper will discuss an analysis of the data access pattern for the D0 experiment and the tools used to perform such a study. 

1. Introduction

To study the fundamental building blocks of matter, the Fermi National Accelerator Laboratory runs several high-energy physics experiments.  One of the experiments, the D0 experiment, observes proton-antiproton collisions at 2TeV(Tera electron volts).  The D0 experiment generates Terabytes of data files used by more than 500 HEP (High Energy Physics) scientists worldwide from numerous universities and laboratories. The SAM (Sequential Access via Meta-data) system is responsible for storing and delivering the data files to physicists.  Registered users can input dataset definitions via the web 

to find data files that match search parameters. One of the goals of the SAM system is to improve the data storage usage. The cache space is limited, and there are many users retrieving data at the same time, so the cache must be organized. Therefore it is important to study the access patterns of the users to the data, for the SAM administrators to have a greater insight on management decisions such as, how much cache is needed.

2. Study of Data Access Patterns for the D0 experiment

A SAM data set definition is a user-defined set of criteria that will be processed to produce a set of files and events. When the group or user feels a data set is in the final state of definition, the user can run an analysis program on the collection of files created using the SAM system, defining an analysis project. A running date and a project ID characterize an analysis project among other parameters. The data files located within the analysis projects are of different data types (data tiers), for example, reconstructed, raw, digitized, special.  Pairs of analysis projects running on different dates can share the same data files.   The objective of this particular study is to observe the amount of shared data files in pairs of analysis projects over intervals of time for a given data tier.  The study surveys Analysis Projects (AP) with project ids from APlo-APmid until APhi, where APhi is the value of the project id of the most current Analysis Project, and APlo is the value of the lowest project id from any given interval of days preceding the date of APhi (e.g. 30, 90, 180, 365 days). APmid is the value of APlo plus the user defined range, it is used to calculate the average.

3. Plots and Tables of Data Access Patterns

The two plots produced show  (top) the average percentage of shared data files in a pair of analysis projects vs. the interval of time between the two projects and (bottom) the total amount of project pairs considered vs. their time interval. The plot (top) takes the first analysis project (APlo) and compares its files with the files of each other analysis project that was run 0,1,2…n days apart. The same thing is done for the second analysis project, the third up to the maximum range (e.g. APs 11963-11973 with range of ten, in the top plot).  The results are then averaged to increase the significance of the statistics. 

Figure 1. [Top plot]  Percentage of shared data files in paired analysis projects vs. interval of time between pair of analysis projects for reconstructed data type for the past 365 days [Bottom plot] Total amount of analysis project pairs considered in a given interval of time.

Figure 2. [Top plot] Percentage of shared data files in pair of analysis projects vs. interval of time between pair of analysis projects for reconstructed data type for the past 90 days [Bottom plot] Total amount of analysis project pairs considered in a given interval of time

The interest in these plots is not in the day-to-day fluctuation of data access, rather over a wider period of time (e.g. months).  Therefore, the plots’ foci are on the average trends that the data points follow.

	
	APlo
	APmid
	APhi

	Project ID
	12059
	12069
	31998

	Date
	03-AUG-00
	03-AUG-00
	03-AUG-01

	Average Size(KB)
	1.11121e+06
	
	


Figure 3. Table of the average size in KB of AP Projects averaged over a range of APlo-APmid project ids, and the dates of the Analysis projects for reconstructed data type from the past 365 days. 
	
	APlo
	APmid
	APhi

	Project ID
	25610
	25620
	31998

	Date
	05-MAY-01
	06-MAY-01
	03-AUG-01

	Average Size(KB)
	1.62457e+07
	
	


Figure 4. Table of the average size in KB of AP Projects averaged over a range of APlo-APmid project ids, and the dates of the Analysis projects for reconstructed data type from the past 90 days. 
In addition to the plots, a table of the average size of the analysis projects, and their dates is considered. This added information provides further information about the analysis projects used in the plots.

4. Implementation of Analysis
The analysis was implemented by building a programming package sam_access_plot. The main script, Monthly.csh invokes the C Shell scripts, run_inter.csh and visualize.csh.  The core of the analysis is done in the C Shell script, run_inter.csh that runs using the following categories if programs (refer to Appendix):  

· C Shell script written to query the data base (Oracle)   

· C Shell script written to perform manipulations on the output.

· C Program to perform mathematical manipulations.

· Kumac (PAW) to create plots

· HTML to create a webpage

In addition, a short study was conducted on the sql queries used to find the dates of the analysis projects and sizes of their files.  Overviews of these studies were published at the websites http://projects.fnal.gov/act/sam_cache/statement.html and http://projects.fnal.gov/act/sam_cache/statement2.html. 

5. Summary

By conducting a data access pattern analysis, the SAM administration is left with valuable information about the data access for the D0 experiment.  The administration can see that the data type, reconstructed shows a trend of 50-60% shared files from August 3, 2000 to August 3, 2001 (Figure 3).  This analysis is done by looking at 100 pairs of analysis projects for any interval of time between the analysis projects. If the SAM administration is interested in viewing a shorter trend, they can look at Figure 2, which models the data access pattern for the past 90 days from May 5, 2001 to August 3, 2001(Figure 4), where it is concluded that for reconstructed data in the past 90 days, less than 10% of the data files are shared while looking at 1000 pairs. Thus the files analyzed in AP range 11963 to 11973 are popular for more than 6 months. This information will be published on the web at a link from the SAM webpage. 

8. Appendix

· The C Shell script, Monthly.csh, that invokes all scripts and programs to build HTML page

#!/bin/csh

source /usr/local/etc/setups.csh

#useful packages

setup ghostview

setup ximagetools

## Internal Configuration

#Defines the directory locations

#parent directory

setenv ABSOLUTE /scratch/4/garzogli

#Temporary directory

setenv ROOT_DIR /scratch/4/garzogli/testdir_root

#HTML output subdirectory

setenv FINAL_SUBDIR testdir_final

#HTML output directory location

setenv FINAL_DIR ${ABSOLUTE}/${FINAL_SUBDIR}

#Output directory name

setenv OUTPUT testdir_output

#Output subdirectory

setenv OUTPUT_SUBDIR ${FINAL_SUBDIR}/${OUTPUT}

#Output directory containing plots

setenv OUTPUT_DIR ${ABSOLUTE}/${OUTPUT_SUBDIR}

#Table directory location

setenv TABLE_DIR /scratch/4/garzogli/testdir_table

#size directory name

setenv SIZE testdir_size

#size report subdirectory

setenv SIZE_RPT_SUBDIR ${FINAL_SUBDIR}/${SIZE}

#size report directory location

setenv SIZE_RPT ${ABSOLUTE}/${SIZE_RPT_SUBDIR}

#past reports directory name

setenv PAST_RPT testdir_pastrpt

#past reports from HTML page subdirectory of FINAL_DIR

setenv PAST_RPT_SUBDIR ${FINAL_SUBDIR}/${PAST_RPT}

#past reports from HTML page directory

setenv PAST_RPT_DIR ${ABSOLUTE}/${PAST_RPT_SUBDIR}

#Image sub directory : pics for the web

setenv IMAGE_SUBDIR images

#Image directory : pics for the web

setenv IMAGE_DIR ${FINAL_DIR}/${IMAGE_SUBDIR}

#==============================================================================================================

#assigns a date to the file

setenv datenum `date +%a_%b_%e_%H:%M`

#assigns a name to the filedate that includes the date

setenv filedate "past_rpts${datenum}.html"

#assigns a name to the file that will output the runtime status of run_inter.csh

setenv runinter_log "${ROOT_DIR}/run_inter.out"

setenv DATA_BASE "sqlplus samread/reader@d0ofint1"

###Output files for the HTML script

#Temporary HTML file name

setenv htmltemp_stm "${FINAL_DIR}/analysis_temp.html"

#HTML file name for current report

setenv html_stm  "${FINAL_DIR}/analysis.html"

#Temporary file name for past reports

setenv html1temp_stm "${FINAL_DIR}/past_temp.html"

#HTML file name for past reports

setenv html1_stm "${FINAL_DIR}/past.html"

setenv htmlsize_stm "${FINAL_DIR}/sizes.html"

#Local Variables

set config_file = "monthly.config"

# Parsing the configuration file in the form:

#     Use # for comment

#     Use the key word "range" to define the APlo-APmid range in Analysis projetcs

#     Use the key word "APlo" multiple times to define the various initial APlo

#      giving a range back in days from the most recent AP (i.e.30, 120,180,365) 

#     Use the key word "no_data_tier"  to remove unwanted data tiers from the

#      analysis (the whole list is read from the db).

# Example:

#  range 10

#  APlo  30

#  APlo  90

#  no_data_tier unknown

#==============================================================================================================

### input range

set range = `grep -v "#" ${config_file} | awk '/range/{print $2}'`

### check if there is only one range key word

if (${#range} =~ 1) then

 set range=${range}

 else

 echo " ${config_file} must have one and only one range value "

 exit(0);

endif

#debug printout 

#echo ${range}

#echo ${#range}

#==============================================================================================================

### input APloDays

set APloDays = `grep -v "#" ${config_file} | awk '/APloDays/{print $2}'`

### check whether there is atleast one APlo key word

if (${#APloDays} < 1) then 

 echo " ${config_file} must have at least one APloDays value "

 exit(0);

endif

#debug printout 

#echo ${APloDays}

#echo ${#APloDays}

#==============================================================================================================

### input no_data_tier

set no_data_tier = `grep -v "#" ${config_file} | awk '/no_data_tier/{print $2}'`

#debug printout 

#echo ${no_data_tier}

#echo ${#no_data_tier}

#==============================================================================================================

#==============================================================================================================

### Definition of the data tiers

set sql_stm = "${ROOT_DIR}/Data_tier.sql"

rm -f ${sql_stm}

set i=0

if ($#no_data_tier == 0) then

   echo "select data_tier from data_tier;"              >>${sql_stm}

else

   echo "select data_tier from data_tier where"         >>${sql_stm}

   foreach nodata ($no_data_tier)

   echo  "data_tier != '$nodata'"                       >>${sql_stm}

  @ i++

      if ($i < $#no_data_tier) then

        echo  " and "                                    >>${sql_stm}

      else                                  

        echo   ";"                                       >>${sql_stm}

      endif

   end

endif

set Data = `cat ${sql_stm} | $DATA_BASE | grep -v -E "(----|DATA_TIER|rows|Copyright|SQL|Oracle8i|JServer|With|Connected)" | grep -v ^$` 

#==============================================================================================================

### Definition of the most recent analysis project

set APhi = `echo "select max(AP1.Project_id) from analysis_project AP1;" | $DATA_BASE | grep -v ^$ |grep -v  -E "(MAX|----|Copyright|SQL|Oracle8i|JServer|With|Connected)"`

#debug printout 

#echo $APhi

#==============================================================================================================

#initializing variables i and APlo (initial AP for any given APloDays day) 

set i=1

set APlo

#Definition of the initial analysis projects depending on the configuration file

foreach APlod ($APloDays)

  #echo $APlod

  set currentAPlo =  `echo "select min(AP2.project_id) from Analysis_Project AP2, Analysis_Project AP1 where AP1.project_id=$APhi and AP2.create_date> (AP1.create_date-$APlod);" | $DATA_BASE | grep -v ^$ |grep -v  -E "(MIN|----|Command|Copyright|SQL|Oracle8i|JServer|With|Connected)"`                                

  set APlo = ($APlo $currentAPlo)

   #debug printout 

   #echo APlos:

   #echo $APlo[$i]

  @ i++

end

#==============================================================================================================

#Definition of the mid analysis projects depending on the APlo

set APmid

foreach L ($APlo)

  set APmd

  @ APmd = $L + $range 

  set APmid = ($APmid $APmd)

end

#==============================================================================================================

#==============================================================================================================

#Plot generation

foreach dt ($Data)

set n=1

  foreach APL ($APlo)

    echo "run_inter.csh ${APL} ${APhi} ${dt} ${range}"             >>${runinter_log}

   run_inter.csh ${APL} ${APhi} ${dt} ${range}

   setenv PLOT "${OUTPUT_DIR}/plot${APL}-${APhi}_${dt}_loc$APmid[$n].jpg"

   size.csh ${APL} ${APhi} ${dt} ${range}

   @ n++

  end

end

#==============================================================================================================

#==============================================================================================================

#HTML generation

#sets environment variable values for visualize.csh 

setenv dat "$Data"

setenv mid "$APmid"

setenv  APLO "$APlo"

setenv APhi "$APhi"

setenv RANGE "$range"

setenv loDays "$APloDays"

visualize.csh  

· Monthly.config text file used within Monthly.csh to define variables.

#Configuration file for monthly.csh

#     Use # for comment

#     Use the key word "range" to define the APlo-APmid range in Analysis projects

#     Use the key word APloDays multiple times to define the various initial APlo

#      giving a range back in days from the most recent AP (i.e.30, 90,180,365) 

#     Use the key word no_data_tier multiple times to define which data_tiers not 

#      to be used in monthly.csh.

range 10

APloDays  30

APloDays  90

APloDays  180

APloDays  365

no_data_tier edu100

no_data_tier edu150

no_data_tier edu50

no_data_tier root-tuple

no_data_tier run_1_sta

no_data_tier edu250

no_data_tier thumbnail

no_data_tier special

#no_data_tier raw

no_data_tier simulated

no_data_tier generated

#no_data_tier digitized

#no_data_tier reconstructed

no_data_tier unknown

no_data_tier epics

no_data_tier run_1_sta

· The C Shell script, run_inter.csh, that contains scripts and program to query, manipulate, and plot data 

#!/bin/csh

if ("$4" != "") then

  set APlo = $1

  set APhi = $2

  set Data = $3

  set range = $4

else

  echo "$0 usage:"

  echo "   $0 APlo APhi Data Range"

  echo "   where the analysis projects considered are from APlo to APlo+range"

  echo "   and the intersection of files is searched through APhi"

  exit(0);

endif

# This information is calculated also in monthly.csh. Passing these values as arguments and

# calculating it again is a redundancy that makes this script independent from the monthly.csh

# "calculated" environment. 

set APmid

@ APmid = $APlo + $range

if($APmid > $APhi) then

  echo "Final $Data analysis project ( $APhi ) smaller than range of interest:"

  echo "from $APlo to $APmid"

  exit

endif

date                                                                      >>${runinter_log}

echo "Searching for intersections on APs ${APlo} - ${APmid} until ${APhi}" >>${runinter_log}

echo "creating table of number of files..."                                >>${runinter_log}

nfiles.csh ${APlo} ${APhi}

ls -la ${TABLE_DIR}/nfiles.csh${APlo}-${APhi}.tbl                          >>${runinter_log}

echo "creating table of days..."                                           >>${runinter_log}

days.csh ${APlo} ${APhi} ${Data} ${range} 

ls -la ${TABLE_DIR}/days.csh${APlo}-${APhi}_${Data}_loc${APmid}.tbl        >>${runinter_log}

echo "creating table of intersecting files..."                             >>${runinter_log}

inter.csh ${APlo} ${APhi} ${Data} ${range}

ls -la ${TABLE_DIR}/inter.csh${APlo}-${APhi}_${Data}_loc${APmid}.tbl       >>${runinter_log}

echo "merging table of days and table of intersecting files..."            >>${runinter_log}

merge.csh ${APlo} ${APhi} ${Data} ${range}

ls -la ${TABLE_DIR}/merge${APlo}-${APhi}_${Data}_loc${APmid}.tbl           >>${runinter_log}

echo "creating histogram data..."                                          >>${runinter_log}

set max_lines = `wc ${TABLE_DIR}/merge${APlo}-${APhi}_${Data}_loc${APmid}.tbl | cut -b -14`

echo "histo ${APlo} ${APhi} ${Data} $range ${max_lines} 200 730"           >>${runinter_log}

histo ${APlo} ${APhi} ${Data} ${range} ${max_lines} 200 730\

      >! ${TABLE_DIR}/histo${APlo}-${APhi}_${Data}_loc${APmid}.tbl        

ls -la ${TABLE_DIR}/histo${APlo}-${APhi}_${Data}_loc${APmid}.tbl           >>${runinter_log}

ls -la ${TABLE_DIR}/table.c${APlo}-${APhi}_${Data}_loc${APmid}.tbl         >>${runinter_log}

echo "DONE"                                                                >>${runinter_log}

date                                                                      >>${runinter_log}

echo "Running paw in batch..."                                             >>${runinter_log}

setenv PAW_ARG "${APlo} ${APhi} ${Data} ${APmid} ${TABLE_DIR} ${OUTPUT_DIR}"

paw -b acc_pat_b_data_loc >&! /dev/null

echo "Gathering info about the analysis projects..."                      >>${runinter_log}

convert ${OUTPUT_DIR}/plot${APlo}-${APhi}_${Data}_loc${APmid}.ps ${OUTPUT_DIR}/plot${APlo}-${APhi}_${Data}_loc${APmid}.jpg

#convert -crop 560x560-15-140 ../../scratch4/testdir_final/testdir_output/plot5260-23873_reconstructed_loc5270.ps test.jpg

#-crop <width>{%}x<height>{%}{+-}<x offset>{+-}<y offset>

#echo "showing plot."

#ghostview ${OUTPUT_DIR}/plot${APlo}-${APhi}_${Data}_loc${APmid}.ps &

echo "$0 DONE"                                                           >>${runinter_log}

echo

#

#

#

#

#

#

#

#

· HTML page for the analysis of the date SQL queries

Analysis of SQL Date Queries
Anne Futrell-2001 SIST Intern

Our goal for this project is to create a programming package that will analyze the data access pattern for the D0 experiments, and provide this information to the public on an updated webpage. For further information on this project visit the fnal projects webpage at projects.fnal.gov. For this analysis we will use sql queries to retrieve information about the data files. Therefore, we would like to find the most expedient and least resource consuming query that outputs the project ids, the dates of the analysis projects, and the difference in the dates between Analysis Project 1 (AP1), and Analysis Project 2 (AP2). The project ids are constrained between the values APlo to APmid and APlo to APhi. We limit the project ids in order to view a smaller segment of the data files rather than the whole. In addition, the data type is defined as reconstructed. We have written 3 different queries that all implement the same task described above.

QUERY1

select distinct AP1.PROJECT_ID, AP2.PROJECT_ID, AP1.CREATE_DATE

from analysis_projects AP1, analysis_projects AP2,   

     project_files PF1, project_files PF2,                 

     data_files DF1, data_files DF2                        

where (PF1.proj_snap_id=AP1.proj_snap_id AND        

     PF2.proj_snap_id=AP2.proj_snap_id AND                

     PF1.FILE_ID = DF1.FILE_ID AND 

     PF2.FILE_ID = DF2.FILE_ID)

     AND (DF1.data_tier = 'reconstructed' AND           

     DF2.data_tier = 'reconstructed') AND                

     (AP1.PROJECT_ID between 9000 and 9010 AND             

     AP2.PROJECT_ID between 9000 and 23000 AND              

     AP2.PROJECT_ID > AP1.PROJECT_ID);     

Query1 does not use any nested SELECT statements. 

The query utilizes joins to connect the tables: analysis_ project, 

project_files, and data_files.
=========================================================================   

QUERY2
select AP.project_id,Q1.project_id,                                          

     AP.create_date from analysis_project AP,                                 

     Project_files PF, data_files DF, 

       (

         select AP2.project_id,AP2.Create_date, 

                PF2.File_id 

         from analysis_project AP2, Project_File PF2, 

         data_files DF2 

         where AP2.Project_id between 9000 and 23000 AND

         DF2.data_tier = 'reconstructed' AND                    

         PF2.Proj_snap_id = AP2.Proj_snap_id AND 

         PF2.File_id=DF2.File_id

       ) Q1      

where AP.project_id between 9000 and 9010 AND 

     DF.data_tier ='reconstructed

     AND AP.proj_snap_id=PF.proj_snap_id                                      

     AND PF.File_id=DF.File_id AND                                         

     AP.project_id < Q1.project_id 

     order by AP.project_id, Q1.project_id ; 

Query2 uses a nested SELECT statement within the

FROM statement of the main SELECT statement. The SELECT statement acts

as a table, which contains the columns project_id, create_date and file_id.
========================================================================= 

QUERY3
select AP1.project_id, AP2.project_id, AP1.create_date          

from analysis_projects AP1, analysis_projects AP2               

where 

     AP1.project_id in 

        (


select analysis_projects.project_id     

        from project_files, analysis_projects, data_files 


where (analysis_projects.proj_snap_id =


project_files.proj_snap_id) AND

        (project_files.file_id = data_files.file_id) AND                

        data_files.data_tier = 'reconstructed' AND                      

        analysis_projects.project_id between 9000 and 


9010) 


 AND        

    AP2.project_id in 

       (

       select analysis_projects.project_id 

       from project_files, analysis_projects, data_files 

       where (analysis_projects.proj_snap_id = 

       project_files.proj_snap_id) AND

       (project_files.file_id = data_files.file_id) AND               

       data_files.data_tier = 'reconstructed' AND                     

       analysis_projects.project_id between 9000 and 

       23000)            

AND AP2.project_id > AP1.project_id;          

Query3 uses two nested SELECT statements in the WHERE statement. 

The SELECT statements in this case are used as control variables for 

AP1.project_id and AP2.project_id.
=============================================================== 

The following timetable gives the system time in seconds for Query1, Query2, and Query3 at different APlo, APmid and APhi values. APlo to APhi gives the whole range, and APlo to APmid gives us a range in which we can take the average over the quantities we look at APlo to APhi.

	Table of time
Query1 Query2 and Query3 

	APlo-APmid-APhi
	Query1
	Query2
	Query3

	19000-19010-20000
	0.826s 
	77.742s
	1.579s 

	19000-19010-21000
	0.932s
	174.883s 
	0.910s

	19000-19010-22000
	1.075s
	215.464s
	0.992s

	10000-10010-13000
	1.126s
	43.244s 
	1.085s 

	10000-10010-14000
	1.248s 
	52.241s 
	1.196s 

	10000-10010-15000
	1.392s 
	70.827s 
	1.451s 

	10000-10010-16000
	1.678s 
	77.926s 
	1.542s 


After analyzing the times, and performing the Oracle's explain statement for the 3 queries, it is clear that Query3 has the shortest running time; the server does not perform any Cartesian joins, which generally increase the amount of resources to use. Therefore we will use Query3 to retrieve the dates, and the difference of the dates for the analysis projects. 

· Monthly.csh User Manual

Overall Function of Programming Package


Sam_access_plots is a programming package that invokes C Shell scripts, C program, and Kumac (PAW) that query the D0 database for information on analysis projects, prints the information onto tables, performs mathematical manipulations, creates histograms, plots data, and publish the results as HTML pages. Monthly.csh carries no command arguments. The HTML pages publish plots to illustrate the data access patterns. The plots produced are of the percentage of shared data files in a pair of analysis projects, between a given distances in time.  And the total amount of analysis project pairs viewed between a given distances in time.  The variables used to create these plots are:

a. APLO- lowest analysis project id which is automatically found from the database, or the most current project,

b. APLODAYS- the number of days in the past the user would like to conduct the study, 

c. RANGE-used to take the average over the analysis projects to increase the significance of the statistics, 

d. APHI- the most recent analysis project 

e. DATA- the type of data file (e.g. raw, reconstructed, special). 

 The user in the configuration file, Monthly.config (refer to page 11 of Appendix), can change the Variables RANGE, APLODAYS and DATA.

Configuration File 


Monthly.config is a configuration file read from Monthly.csh.  This text-only file is included in the directory, src, of the programming package.
 The user inputs key words (e.g. range, AploDays, no_data_tier) to define the variables.  The user can only enter one range value, but multiple values for AploDays and no_data_tier. The range value is used in monthly.csh to define the Aplo- Apmid range in analysis project.  The AploDays is the number of days in the past the user is interested in viewing from the current date.  And, no_data_tiers are the data types that the user wants excluded from viewing the analysis.

Monthly.csh Environment Variables

The environment variables for the locations of the file output directories and subdirectories are specified in Monthly.csh.  The environment variables are: 

· ABSOLUTE- the parent directory 

· ROOT_DIR- the directory that holds temporary files

· TABLE_DIR- the directory that holds all tables from sql queries

· FINAL_DIR- the directory that holds the output of the HTML page

· OUTPUT_DIR- a subdirectory of FINAL_DIR that holds the current HTML page which includes plots and tables

· SIZE_RPT- a subdirectory of FINAL_DIR that holds the page that is linked to each plot that holds the plot and table of size of data files

· PAST_RPT_DIR- subdirectory of FINAL_DIR where the past HTML pages are located

· IMAGE_DIR- subdirectory of FINAL_DIR that holds all images used in HTML page.

· The HTML page of the Analysis of the Data Access Pattern for the D0 experiment

For more information on this project, please visit the fnal projects webpage at projects.fnal.gov.
�  http://d0db.fnal.gov/sam/doc/userdocs/running_sam_user_tips.html

